A Self-Guided Tour of NESL

NESL is a parallel language developed here at Carnegie Mellon. It is loosely based on the programming language ML. This document will help you get started with NESL, which we will use in this class.

1 Loading NESL

NESL has an interactive interface that runs on Andrew Linux machines. To start NESL, type

```
/afs/cs.cmu.edu/project/scandal/nesl/runnesl
```

at a shell prompt. You should have the following screen, ending with a NESL prompt `<Nesl>`.



```
Copyright (c) Bruno Haible, Michael Stoll 1992, 1993
Copyright (c) Bruno Haible, Marcus Daniels 1994-1997
Copyright (c) Bruno Haible, Pierpaolo Bernardi, Sam Steingold 1998
Copyright (c) Bruno Haible, Sam Steingold 1999-2000
Copyright (c) Sam Steingold, Bruno Haible 2001-2006


% NESL version 3.1 (November 1, 1995) %

% Setting machine configuration LOCAL.... %

% Use (nesl) when an error aborts you out of the interpreter. %
% Type help; for a list of the top level commands. %

<Nesl>
```

Now that you can start NESL, an important command to learn is `exit;`. This will bring you back to the shell. You can also exit NESL with Ctrl-D.

In the remaining of the document, you will learn to operate NESL by examples. For a quick-reference guide, visit

```
http://www.cs.cmu.edu/~scandal/nesl/quickref.html
```

2 Getting Help

```
<Nesl> describe ++;
```

```
INTERFACE:

v1 ++ v2

TYPE:

 ([a], [a]) -> [a] :: (a in any)

DOCUMENTATION:

 Given two sequences, {\tt ++} appends them.

CODE:

  join(v1,
       iseq-l(0, 1, v1), v2, iseq-l(length(v1), 1, v2))


<Nesl> help();

NESL top-level forms:
  function <name> <pattern> [: <typedef>] = <exp>;  -- Function Definition
  datatype <name> <typedef>;                        -- Record Definition
  <pattern> = <exp>;                                -- Top level Assignment
  <exp>;                                            -- Any NESL expression

 .
 .
 .
```

3 Scalar Operations

```
<Nesl>

it = 14 : int
<Nesl> (2.2 + 1.1) / 5.0;

it = 0.66 : float
<Nesl> T or F;

it = T : bool
<Nesl> ‘a < ‘d;

it = T : bool
<Nesl> 1.6 + 7;

Error at top level.
For function + in expression
  1.6 + 7
inferred argument types don’t match function specification.
  Argument types: float, int
```

```
  Function types: a, a :: a in number

<Nesl> it = T : bool

it = 8.6 : float
<Nesl> round(1.6) + 7;

it = 9 : int
<Nesl> sin(.6);

it = 0.564642473395035 : float
<Nesl> a = 4;

a = 4 : int
<Nesl> a + 5;

it = 9 : int
<Nesl> if (4 < 5) then 11 else 12;

it = 11 : int
<Nesl> let a = 3 * 4
>      in a + (a * 1);

it = 24 : int
<Nesl> let a = 3 * 4;
>          b = a + 5
>      in a + b;

it = 29 : int
<Nesl> function fact(i) =
>        if (i == 1)
>        then 1
>        else i * (fact(i-1));

fact = fn : int -> int
<Nesl> fact(5);

it = 120 : int
<Nesl> function circarea(r) = pi * r * r;

circarea = fn : float -> float
<Nesl> circarea(3.0);

it = 28.2743338823081 : float
<Nesl> (2, `a);

it = (2, `a) : int, char
<Nesl> function div_rem(a, b) = (a / b, rem(a, b));

div_rem = fn : (int, int) -> (int, int)
<Nesl> div_rem (20, 6);
```

```
it = (3, 2) : int, int
```

4 Vector Operations

```
<Nesl> [2, 5, 1, 3];

it = [2, 5, 1, 3] : [int]
<Nesl> "this is a vector";

it = "this is a vector" : [char]
<Nesl> [(2, 3.4), (5, 8.9)];

it = [(2, 3.4), (5, 8.9)] : [(int, float)]
<Nesl> [2, 3.0, 4];

Error at top level.
For function make_sequence in expression
  [2, 3.0]
inferred argument types don't match function specification.
  Argument types: [int], float
  Function types: [a], a :: (a in any)

<Nesl> {a + 1: a in [2, 3, 4]};

it = [3, 4, 5] : [int]
<Nesl> let a = [2, 3, 4] in {a + 1: a};

it = [3, 4, 5] : [int]
<Nesl> {a + b: a in [2, 3, 4]; b in [4, 5, 6]};

it = [6, 8, 10] : [int]
<Nesl> let a = [2, 3, 4]; b = [4, 5, 6] in {a + b: a; b};

it = [6, 8, 10] : [int]
<Nesl> {a == b: a in "this"; b in "that"};

it = [T, T, F, F] : [bool]
<Nesl> {fact(a): a in [1, 2, 3, 4, 5]};

it = [1, 2, 6, 24, 120] : [int]
<Nesl> {div_rem(100, a): a in [5, 6, 7, 8]};

it = [(20, 0), (16, 4), (14, 2), (12, 4)] : [(int, int)]
<Nesl> sum([2, 3, 4]);

it = 9 : int
<Nesl> dist(5, 10);

it = [5, 5, 5, 5, 5, 5, 5, 5, 5, 5] : [int]
<Nesl> [2:50:3];
```

```
it = [2, 5, 8, 11, 14, 17, 20, 23, 26, 29, 32, 35, 38, 41, 44, 47] : [int]
<Nesl> "big" ++ " boy";

it = "big boy" : [char]
<Nesl> {x in "wombat" | x <= `m};

it = "mba" : [char]
<Nesl> {sum(a): a in [[2, 3, 4], [1], [7, 8, 9]]};

it = [9, 1, 24] : [int]
<Nesl> bottop("testing");

it = ["test", "ing"] : [[char]]
<Nesl> partition("break into words", [5, 5, 6]);

it = ["break", " into", " words"] : [[char]]
<Nesl> function my_sum(a) =
>         if (#a == 1) then a[0]
>         else
>           let res = {my_sum(x): x in bottop(a)}
>           in res[0] + res[1];

my_sum = fn : [a] -> a :: (a in number)
<Nesl> my_sum([7, 2, 6]);

it = 15 : int
```

5 Loading Program Files

```
<Nesl> load("/afs/cs/academic/class/15492-f07/nesl/examples/median.nesl");

% Loading /afs/cs/academic/class/15492-f07/nesl/examples/median.nesl. %

<Nesl> my_median([8, 1, 2, 9, 5, 3, 4]);

it = 4 : int
<Nesl> my_median("this is a test");

it = `i : char
```